// File: contracts/GarbageVesting.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.18;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "./IGarbageVesting.sol";

contract GarbageVesting is Ownable, IGarbageVesting {
    using SafeERC20 for IERC20;
    struct Beneficiary {
        uint256 amount;
        uint256 vestingAmount;
        uint256 claimed;
    }

    // _______________ Storage _______________

    /// @dev Number of vesting periods.
    uint256 public vestingPeriods;

    /// @dev Vesting period duration.
    uint256 public vestingPeriodDuration;

    /// @dev Vesting start timestamp.
    uint256 public startTimestamp;

    /// @dev Vesting end timestamp.
    uint256 public endTimestamp;

    /// @dev Sum of all vesting amounts added to the contract.
    uint256 public totalVestingAmount;

    /// @dev vestingToken $GARBAGE token address.
    IERC20 public vestingToken;

    /// @dev Garbage sale contract address.
    address public garbageSale;

    // beneficiary address => beneficiary data struct
    mapping(address => Beneficiary) public beneficiaries;

    // amount of tokens claimed by all beneficiaries
    uint256 public totalClaimed;

    uint256 public constant VESTING_DURATION = 10 weeks;

    // _______________ Errors _______________

    /// @dev Revert is zero address is passed.
    error ZeroAddress();

    /// @dev Revert if zero vesting amount is passed.
    error ZeroVestingAmount(address _beneficiary);

    error LessThanTwoVestingPeriods(uint256 _vestingPeriods);

    /// @dev Revert if beneficiary does not exist.
    error NotABeneficiary(address _user);

    /// @dev Revert if caller is not a garbage sale contract.
    error NotAGarbageSale(address _caller);

    /// @dev Revert if vesting is not initialized.
    error VestingDatesNotSet();

    /// @dev Revert if vesting period is not set.
    error VestingPeriodNotSet();

    /// @dev Revert if user has already claimed.
    error NothingToClaim(address user);

    /// @dev Revert when trying to rescue vesting tokens.
    error VestingTokenRescue(IERC20 vestingToken);

    /// @dev Revert if end timestamp is before start timestamp.
    error EndTimestampBeforeStartTimestamp(uint256 end, uint256 start);

    /// @dev Revert if claim amount exceeds vesting amount.
    error ClaimAmountExceedsVestingAmount(address _beneficiary, uint256 _claimAmount, uint256 _amount);

    /// @dev Revert if start timestamp is before current timestamp (now).
    error StartTimestampBeforeCurrentTimestamp(uint256 start, uint256 current);

    /// @dev Revert if vesting already started.
    error VestingAlreadyStarted();

    // _______________ Events _______________

    /**
     * @dev Emitted when the claim is successful.
     *
     * @param _user   Address of the user.
     * @param _amount   Amount of the claim.
     */
    event Claim(address indexed _user, uint256 indexed _amount);

    /**
     * @dev Emitted when a new beneficiary is added or when additional amount is added to the existing beneficiary.
     *
     * @param _beneficiary   Address of the beneficiary.
     * @param _newAmount   Total amount to vest to the user.
     * @param _newVestingAmount   Token amount per one vesting period.
     */
    event BeneficiaryUpdated(address indexed _beneficiary, uint256 indexed _newAmount, uint256 _newVestingAmount);

    /**
     * @dev Emitted when the start and/or end date is updated.
     */
    event VestingDatesUpdated(uint256 indexed _startTimestamp, uint256 indexed _endTimestamp);

    /**
     * @dev Emitted when the vestingToken token address is updated.
     */
    event VestingTokenUpdated(address indexed _vestingToken);

    /**
     * @dev Emitted when the garbage sale contract address is updated.
     */
    event GarbageVestingUpdated(address indexed _garbageVesting);

    /**
     * @dev Emitted when the number of vesting periods is updated.
     */
    event VestingPeriodsNumberUpdated(uint256 _vestingPeriodsNumber);

    /**
     * @dev Emitted when ERC20 tokens are rescued.
     */
    event ERC20Rescued(address indexed _token, address indexed _to, uint256 indexed _amount);

    // _______________ Modifiers _______________

    /**
     * @dev Zero address check.
     */
    modifier notZeroAddress(address _address) {
        if (_address == address(0)) {
            revert ZeroAddress();
        }
        _;
    }

    /**
     * @dev Check if _user is a beneficiary.
     */
    modifier onlyBeneficiary(address _user) {
        if (beneficiaries[_user].amount == 0) {
            revert NotABeneficiary(_user);
        }
        _;
    }

    /**
     * @dev Check if mgs.sender is a garbage sale contract.
     */
    modifier onlyGarbageSale() {
        if (msg.sender != garbageSale) {
            revert NotAGarbageSale(msg.sender);
        }
        _;
    }

    // _______________ Constructor ______________

    /**
     * @dev Initialize vesting.
     *
     * @param _startTimestamp   Start timestamp of the vesting.
     * @param _vestingToken   Address of the vestingToken token.
     * @param _vestingPeriods   Number of vesting periods.
     */
    constructor(
        IERC20 _vestingToken,
        uint256 _startTimestamp,
        uint256 _vestingPeriods
    ) Ownable() {
        if (_startTimestamp <= block.timestamp)
            revert StartTimestampBeforeCurrentTimestamp(_startTimestamp, block.timestamp);
        // init variables
        _setVestingStartDate(_startTimestamp);
        _setVestingToken(_vestingToken);
        _setVestingPeriods(_vestingPeriods);
    }

    // _______________ External functions _______________

    /**
     * @dev Claim tokens.
     */
    function claim() external onlyBeneficiary(msg.sender) {
        uint256 amount = calculateClaimAmount(msg.sender);
        if (amount == 0) {
            revert NothingToClaim(msg.sender);
        }
        beneficiaries[msg.sender].claimed += amount;
        totalClaimed += amount;

        vestingToken.safeTransfer(msg.sender, amount);

        emit Claim(msg.sender, amount);
    }

    /**
     * @dev Update start timestamp. End timestamp is calculated based on the start timestamp.
     * @param _startTimestamp  Start timestamp of the vesting.
     */
    function setStartTimestamp(uint256 _startTimestamp) external onlyOwner {
        _setVestingStartDate(_startTimestamp);
    }

    /**
     * @dev Set new GarbageSale contract address.
     * @param _garbageSale Address of the new GarbageSale contract.
     */
    function setGarbageSale(address _garbageSale) external onlyOwner {
        _setGarbageSale(_garbageSale);
    }

    /**
     * @dev Rescue ERC20 tokens from the contract. Token must be not vestingToken.
     * @param _token Address of the token to rescue.
     * @param _to Address to send tokens to.
     */
    function rescueERC20(IERC20 _token, address _to)
        external
        onlyOwner
        notZeroAddress(address(_token))
        notZeroAddress(_to)
    {
        if (vestingToken == _token) {
            revert VestingTokenRescue(vestingToken);
        }
        emit ERC20Rescued(address(_token), _to, _token.balanceOf(address(this)));
        _token.safeTransfer(_to, _token.balanceOf(address(this)));
    }

    // _______________ Garbage sale functions _______________

    /**
     * @dev Allows GarbageSale contract to add beneficiary. If beneficiary already exists, it adds amount to the existing beneficiary.
     * @param _beneficiary Address of the beneficiary.
     * @param _amount Amount of tokens to be vested.
     */
    function addAmountToBeneficiary(address _beneficiary, uint256 _amount) external onlyGarbageSale {
        vestingToken.safeTransferFrom(msg.sender, address(this), _amount);
        _addAmountToBeneficiary(_beneficiary, _amount);
    }

    // _______________ Public functions _______________

    /**
     * @dev Calculate claim amount for the beneficiary. It returns 0 if vesting is not active or beneficiary has already claimed all tokens.
     *
     * @param _beneficiary   Address of the beneficiary.
     * @return claimAmount   Amount of tokens that can be claimed by the beneficiary.
     */
    function calculateClaimAmount(address _beneficiary)
        public
        view
        onlyBeneficiary(_beneficiary)
        returns (uint256 claimAmount)
    {
        Beneficiary storage beneficiary = beneficiaries[_beneficiary];

        // theoretically should never happen, but just in case
        if (startTimestamp == 0 || endTimestamp == 0) return 0;
        if (beneficiary.amount == 0 || beneficiary.amount <= beneficiary.claimed) return 0;
        if (block.timestamp <= startTimestamp) return 0;

        uint256 vestedPeriods = (block.timestamp - startTimestamp) / vestingPeriodDuration;
        // because we already checked that block.timestamp > initialUnlockTimestamp, so user can claim at least initialAmount
        uint256 claimableAmount = 0;

        if (vestedPeriods >= vestingPeriods) {
            claimableAmount = beneficiary.amount;
        } else {
            claimableAmount += vestedPeriods * beneficiary.vestingAmount;
        }
        claimAmount = claimableAmount - beneficiary.claimed;
        // should never happen, but just in case
        if (claimAmount + beneficiary.claimed > beneficiary.amount)
            revert ClaimAmountExceedsVestingAmount(_beneficiary, claimAmount, beneficiary.amount);
    }

    /**
     * @dev Returns all vesting data.
     *
     * @return _totalVestingAmount   Total amount of vesting tokens.
     * @return _startTimestamp   Start timestamp of the vesting.
     * @return _endTimestamp   End timestamp of the vesting.
     * @return _vestingPeriods   Number of vesting periods.
     * @return _vestingPeriodDuration   Duration of the vesting period.
     * @return _vestingToken   Address of the vestingToken token.
     */
    function getAllVestingData()
        external
        view
        returns (
            uint256 _totalVestingAmount,
            uint256 _startTimestamp,
            uint256 _endTimestamp,
            uint256 _vestingPeriods,
            uint256 _vestingPeriodDuration,
            address _vestingToken
        )
    {
        _totalVestingAmount = totalVestingAmount;
        _startTimestamp = startTimestamp;
        _endTimestamp = endTimestamp;
        _vestingPeriods = vestingPeriods;
        _vestingPeriodDuration = vestingPeriodDuration;
        _vestingToken = address(vestingToken);
    }

    // _______________ Internal functions _______________

    /**
     * @dev Add beneficiary to the vesting.
     *
     * @param _beneficiary   Address of the beneficiary.
     * @param _amount   Amount of tokens to be vested.
     */
    function _addAmountToBeneficiary(address _beneficiary, uint256 _amount) internal notZeroAddress(_beneficiary) {
        // theoretically it should never happen, but just in case
        if (vestingPeriods == 0) revert VestingPeriodNotSet();
        if (_amount == 0) revert ZeroVestingAmount(_beneficiary);

        beneficiaries[_beneficiary].amount += _amount;
        beneficiaries[_beneficiary].vestingAmount = beneficiaries[_beneficiary].amount / vestingPeriods;

        emit BeneficiaryUpdated(
            _beneficiary,
            beneficiaries[_beneficiary].amount,
            beneficiaries[_beneficiary].vestingAmount
        );
        totalVestingAmount = totalVestingAmount + _amount;
    }

    /**
     * @dev Set the claim start timestamp. End timestamp is calculated based on the start timestamp.
     *
     * @param _startTimestamp Claim start timestamp.
     */
    function _setVestingStartDate(uint256 _startTimestamp) internal {
        if (block.timestamp > startTimestamp && startTimestamp != 0) revert VestingAlreadyStarted();
        startTimestamp = _startTimestamp;
        endTimestamp = startTimestamp + VESTING_DURATION;
        emit VestingDatesUpdated(_startTimestamp, endTimestamp);
    }

    /**
     * @dev Set the vestingToken token address.
     * Requirements:
     * - vestingToken token address must not be zero address.
     * @param _vestingToken vestingToken token address.
     */
    function _setVestingToken(IERC20 _vestingToken) internal notZeroAddress(address(_vestingToken)) {
        vestingToken = _vestingToken;
        emit VestingTokenUpdated(address(_vestingToken));
    }

    /**
     * @dev Set address of the garbage sale contract.
     * Requirements:
     * - garbage sale contract address must not be zero address.
     * @param _garbageSale Address of the garbage sale contract.
     */
    function _setGarbageSale(address _garbageSale) internal notZeroAddress(_garbageSale) {
        garbageSale = _garbageSale;
        emit GarbageVestingUpdated(_garbageSale);
    }

    /**
     * @dev Set the vesting periods.
     * Requirements:
     * - vesting periods must be greater than 1.
     * @param _vestingPeriods Number of vesting periods.
     */
    function _setVestingPeriods(uint256 _vestingPeriods) internal {
        if (_vestingPeriods < 2) revert LessThanTwoVestingPeriods(_vestingPeriods);
        // theoretically it should never happen, but just in case
        if (endTimestamp == 0 || startTimestamp == 0) revert VestingDatesNotSet();
        // theoretically it should never happen, but just in case
        if (endTimestamp <= startTimestamp) revert EndTimestampBeforeStartTimestamp(endTimestamp, startTimestamp);
        vestingPeriods = _vestingPeriods;
        vestingPeriodDuration = (endTimestamp - startTimestamp) / vestingPeriods;
        emit VestingPeriodsNumberUpdated(_vestingPeriods);
    }
}

// File: contracts/IGarbageVesting.sol



pragma solidity 0.8.18;

interface IGarbageVesting {
    function addAmountToBeneficiary(address _beneficiary, uint256 _amount) external;
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/draft-IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/draft-IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly
                /// @solidity memory-safe-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}